# APPENDIX 1
# FEM1

```
C ********************************************************************
C
C                         PROGRAM : FEM1
C
C                          SPRING  1984
C
C ********************************************************************
C
C   PURPOSE:
C       PROGRAM FOR SOLVING HEAT CONDUCTION PROBLEMS USING
C       THREE-NODE LINEAR TRIANGULAR ELEMENTS
C
C   BOUNDARY VALUE PROBLEMS SOLVED:
C
C               - DIV ( K GRAD U ) = F      IN  D
C                              U   = G      ON S1
C                       K GRAD U.N = H      ON S2
C               K GRAD U.N = - A( U - UO )  ON S3
C
C                 K   = HEAT CONDUCTIVITY
C                 U   = TEMPERATURE
C                 F   = HEAT SUPPLY
C                 N   = UNIT VECTOR OUTWARD NORMAL TO THE BOUNDARY
C                 G   = SPECIFIED BOUNDARY TEMPERATURE
C                 H   = SPECIFIED HEAT FLUX THROUGH THE BOUNDARY
C                 UO  = OUTSIDE TEMPERATURE
C                 A   = MATERIAL CONSTANT
C
C   NOTE:
C       THE PROGRAM ASSUMES THE DOUBLE PRECESSION: IMPLICIT REAL*8
C       (A-H,O-Z).
C       THE GLOBAL STIFFNESS MATRIX IS STORED IN THE ONE
C       DIMENSIONAL ARRAY USING THE INDEX TRANSFORMATION.
C       SK(I,J) --> SK(I,J-I+1) --> SK(MBAND*(I-1)+J-I+1)  FOR I<=J
C
C   CONTROL NUMBERS:
C       LBOUN1 = 1  IF THE 1-ST BOUNDARY CONDITION IS ASSUMED
C       LBOUN2 = 1  IF THE 2-ND BOUNDARY CONDITION IS ASSUMED
C       LBOUN3 = 1  IF THE 3-RD BOUNDARY CONDITION IS ASSUMED
C       LHEATG = 1  IF THE HEAT SUPPLY IS ASSUMED
C
C   STRUCTURE OF THE PROGRAM:
C       PRE PROCESSING
C          SUBROUTINE PTITLE  --------------- READ THE TITLE OF THE
C                                             PROBLEM
C          MAIN PROGRAM --------------------- READ CONTROL NUMBERS
C          SUBROUTINE HRMESH  --------------- READ COORDINATES AND
C                                             ELEMENT CONNECTIVITIES
C          SUBROUTINE MATERL  --------------- READ CONDUCTIVITIES
C          SUBROUTINE HBANDW  --------------- COMPUTE HALF-BAND WIDTH
C                                             OF STIFFNESS MATRIX
C       FINITE ELEMENT PROCESSING
C          SUBROUTINE ASEMBO  --------------- ASSEMBLE ELEMENT
C                                             STIFFNESS MATRICES
C                                             AND LOAD VECTORS
C              SUBROUTINE ESTIF3  ----------- ELEMENT STIFFNESS
C                                             MATRIX
C                   SUBROUTINE GRADI3  ------ GRADIENT VECTOR OF
C                                             SHAPE FUNCTIONS
C          SUBROUTINE HEATGE ---------------- HEAT SOURCES/SINKS
C                                             ARE INPUT IN THE LOAD
C                                             VECTOR
C          SUBROUTINE BOUND1  --------------- 1-ST BOUNDARY CONDITION
C          SUBROUTINE BOUND2  --------------- 2-ND BOUNDARY CONDITION
C          SUBROUTINE BOUND3  --------------- 3-RD BOUNDARY CONDITION
C          SUBROUTINE BANDSL  --------------- SOLVES THE SYSTEM OF
C                                             LINEAR EQUATIONS
C       POST PROCESSING
C          SUBROUTINE OUTPUT  --------------- OUTPUT THE RESULTS
C          SUBROUTINE QUALIT ---------------- COMPUTE QUALITY INDICES
C                                             TO CHECK THE RESULTS
C   INPUT DATA TO BE READ:
C       1) SUBROUTINE PTITLE
C            1.0 FLAG(I),I=1,10               :    10A8
C            1.1 TITLE(I),I=1,10              :    10A8
```

```
C       2) MAIN : CONTROL NUMBERS
C            2.0 FLAG(I),I=1,10                  :   10A8
C            2.1 LBOUN1,REM(I),I=1,9             :   I5,9A8
C            2.2 LBOUN2,REM(I),I=1,9             :   I5,9A8
C            2.3 LBOUN3,REM(I),I=1,9             :   I5,9A8
C            2.4 LHEATG,REM(I),I=1,9             :   I5,9A8
C       3) SUBROUTINE HRMESH
C            3.0 FLAG(I),I=1,10                  :   10A8
C            3.1 NODADD,NELADD                   :   2I5
C            3.2 II,X(II),Y(II),I=1,NODADD       :   I5,2F10.4
C            3.3 NN,MPE(NN),(IJK(J,NN),J=1,3),
C                N=1,NELADD                      :   5I5
C       4) SUBROUTINE MATERL
C            4.0 FLAG(I),I=1,10                  :   10A8
C            4.1 MATX                            :   16I5
C            4.2 I,(XYK(J,I),J=1,3),I=1,MATX     :   I5,3F10.4
C       5) SUBROUTINE HEATGE  IF  LHEATG =1
C            5.0 FLAG(I),I=1,10                  :   10A8
C            5.1 NHEATG                          :   16I5
C            IF(NHEATG.LT.0) GO TO 5.3
C            DO 100 I=1,NHEATG
C            5.2 CXY(I,J),J=1,NHEATG             :   8F10.4
C            100 CONTINUE
C            GO TO 7)
C            5.3 NODE,VALUE,I=1,-NHEATG          :   I5,F10.4
C       6) SUBROUTINE BOUND1  IF  LBOUN1 =1
C            6.0 FLAG(I),I=1,10                  :   10A8
C            6.1 NBX,PENALTY                     :   I5,F20.5
C            6.2 NODE,VALUE,I=1,NBX              :   I5,F10.4
C       7) SUBROUTINE BOUND2  IF  LBOUN2 =1
C            7.0 FLAG(I),I=1,10                  :   10A8
C            7.1 NBX                             :   16I5
C            7.2 NODE1,NODE2,VALUE,I=1,NBX       :   2I5,F10.4
C       8) SUBROUTINE BOUND3  IF  LBOUN3 =1
C            8.0 FLAG(I),I=1,10                  :   10A8
C            8.1 NBLX,MPBX                       :   16I5
C            8.2 I,(IJB(J,I),J=1,2),MPB(I),
C                I=1,NBLX                        :   4I5
C            8.3 I,XKB(I),UO(I),I=1,MPBX         :   I5,2F10.4
C
C       *** LOOK AT EACH SUBROUTINE BEFORE MAKING INPUT DATA
C
C   DIMENSIONING:
C       SK(NX*MBAND),F(NX),U(NX),X(NX),Y(NX),CQX(NX),CQY(NX)
C          NX    = TOTAL NUMBER OF NODAL POINTS IN THE DOMAIN
C          MBAND = (HALF-)BAND WIDTH OF THE STIFFNESS MATRIX
C          SK    = BANDED STIFFNESS MATRIX
C          F     = GENERALIZED LOAD VECTOR
C          U     = SOLUTION VECTOR
C          X,Y   = COORDINATES OF NODAL POINTS
C          CQX   = QUALITY INDEX AT NODAL POINTS (TEMPERATURE)
C          CQY   = QUALITY INDEX AT NODAL POINTS (HEAT FLUX)
C       IJK(NODE,NELX),MP(NELX)
C          NELX  = TOTAL NUMBER OF FINITE ELEMENTS
C          NODE  = TOTAL NUMBER OF NODAL POINTS IN AN ELEMENT
C          IJK   = ELEMENT CONNECTIVITIES
C          MP    = MATERIAL GROUP OF THE FINITE ELEMENTS
C       XYK(3,MPX)
C          MPX   = TOTAL NUMBER OF MATERIAL GROUPS
C          XYK   = MATERIAL PROPERTIES (KXX,KYY,KXY)
C       IJB(NODE3,NBX),MPB(NBX),XKB(MBX),UO(MBX)
C          NODE3 = TOTAL NUMBER OF NODAL POINTS ON A LINE ELEMENT
C          NBX   = TOTAL NUMBER OF LINE ELEMENTS
C          MBX   = TOTAL NUMBER OF DIFFERENT MATERIAL GROUPS ALONG
C                  THE 3-RD BOUNDARY
C          IJB   = ELEMENT CONNECTIVITIES ON THE 3-RD BOUNDARY
C          MPB   = GROUP NUMBER OF MATERIALS
C          XKB   = COEFFIENTS OF HEAT TRANSFER
C          UO    = OUTSIDE TEMPERATURE SPECIFIED
C       P(4,NELX)
C          P     = STORES THE HEAT FLUX VECTORS AND QUALITY INDICES
C       FLAG(10),REM(9)
C          FLAG  = DESCRIPTION OF THE DATA BLOCKS
C                  ( ANY COMMENTS WITHIN 10A8 CAN BE INPUT IN ORDER TO
C                  MEMORIZE THE INPUT DATA IN EACH SUBROUTINE.  IF YOU
C                  DO NOT WANT IT, THEN SKIP ONE LINE IN THE DATA FILE )
C          REM   = REMARK TO THE INPUT DATA
C
C   DATA STATEMENT: < IMPORTANT >
```

```
C        LR      = DISK NUMBER FOR READ STATEMENTS  ( STD IS 5 )
C        LW      = DISK NUMBER FOR WRITE STATEMENTS ( STD IS 6 )
C        LP      = DISK NUMBER FOR WRITE STATEMENTS FOR PLOTTING
C                    ( STD IS 8 )
C        NDIM    > MBAND*NX*NDF
C        NODE    = TOTAL NUMBER OF NODAL POINTS IN AN ELEMENT ( = 3 )
C        NDF     = NUMBER OF DEGREES OF FREEDOM PER NODAL POINT ( = 1 )
C
C *********************************************************************
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION SK(50000),F(300),U(300),X(300),Y(300),
     1          IJK(3,500),MP(500),
     3          XYK(3,10),
     4          IJB(2,100),MPB(100),XKB(30),UO(30),
     5          P(4,500),CQX(300),CQY(300),
     6          FLAG(10),REM(10)
      DATA LR,LW,LP,NDIM,NODE,NDF/5,6,8,50000,3,1/
C
C ******************    PRE-PROCESSING   **********************
C
C     ( READ THE TITLE OF THE PROBLEM AND CONTROL NUMBERS )
C
            CALL PTITLE(LR,LW)
C
      READ(LR,502) (FLAG(I),I=1,10)
  502 FORMAT(10A8)
      WRITE(LW,602)
  602 FORMAT(/////10X,'XXXXXXXXXX CONTROL NUMBERS XXXXXXXXXX',//)
  500 FORMAT(I5,9A8)
  600 FORMAT(/10X,I3,3X,9A8)
      READ(LR,500)  LBOUN1,(REM(I),I=1,9)
      WRITE(LW,600) LBOUN1,(REM(I),I=1,9)
      READ(LR,500)  LBOUN2,(REM(I),I=1,9)
      WRITE(LW,600) LBOUN2,(REM(I),I=1,9)
      READ(LR,500)  LBOUN3,(REM(I),I=1,9)
      WRITE(LW,600) LBOUN3,(REM(I),I=1,9)
      READ(LR,500)  LHEATG,(REM(I),I=1,9)
      WRITE(LW,600) LHEATG,(REM(I),I=1,9)
C
C     ( READ THE FINITE ELEMENT MODEL )
C
            CALL HRMESH(NX,NELX,X,Y,IJK,NODE,MP,LR,LW)
C
C     ( READ THE MATERIAL PROPERTIES )
C
            CALL MATERL(XYK,LR,LW)
C
C     ( FIND THE HALF-BAND WIDTH OF THE STIFFNESS MATRIX )
C
            CALL HBANDW(MBAND,IJK,NELX,NODE,NDF,LW)
C
      IF(NX*MBAND.LE.NDIM) GO TO 100
      WRITE(LW,644)
  644 FORMAT(/////,'ARRAY FOR THE STIFFNESS MATRIX IS TOO SMALL')
      STOP
  100 CONTINUE
C
C *************   FINITE ELEMENT-PROCESSING   ****************
C
C     ( FORM THE STIFFNESS MATRIX AND THE LOAD VECTOR )
C
            CALL ASEMBO(SK,F,MBAND,NX,NELX,IJK,NODE,NDF,MP,X,Y,XYK,LW)
C
C     ( ROUTINE FOR THE HEAT SUPPLY TERM )
C
      IF(LHEATG.EQ.1) CALL HEATGE(F,NX,NELX,IJK,X,Y,LR,LW)
C
C     ( ROUTINES FOR THE BOUNDARY CONDITIONS )
C
      IF(LBOUN1.EQ.1) CALL BOUND1(MBAND,SK,F,LR,LW)
      IF(LBOUN2.EQ.1) CALL BOUND2(F,X,Y,LR,LW)
      IF(LBOUN3.EQ.1) CALL BOUND3(MBAND,SK,F,X,Y,IJB,MPB,XKB,UO,LR,LW)
C
C     ( SOLVE THE SYSTEM OF LINEAR EQUATIONS )
C
            CALL BANDSL(SK,F,NX,MBAND,0)
C
```

```
C ****************   POST-PROCESSING   *********************
C
C     ( PRINT OUT THE RESULTS )
C
            CALL OUTPUT(F,P,NX,NELX,X,Y,IJK,MP,XYK,LW)
C
            CALL QUALIT(NELX,NX,MBAND,IJK,X,Y,P,SK,CQX,CQY,NODE,LW)
C
C     ( STORE THE RESULTS FOR THE PLOTTING ROUTINE )
C
      WRITE(LP,810)
  810 FORMAT('FINITE ELEMENT MODEL')
      WRITE(LP,800) NX,NELX
  800 FORMAT(2I5)
      WRITE(LP,820)
  820 FORMAT(' < NODAL COORDINATES >')
      WRITE(LP,802) (X(I),Y(I),I=1,NX)
  802 FORMAT(8F10.4)
      WRITE(LP,812)
  812 FORMAT(' < ELEMENT CONNECTIVITIES >')
      WRITE(LP,804) ((IJK(J,I),J=1,NODE),I=1,NELX)
  804 FORMAT(16I5)
      WRITE(LP,814)
  814 FORMAT('TEMPERATURE')
      WRITE(LP,806) (F(I),I=1,NX)
  806 FORMAT(8E10.3)
      WRITE(LP,816)
  816 FORMAT('HEAT FLUX AND QUALITY INDICES')
      WRITE(LP,808) ((P(J,I),J=1,4),I=1,NELX)
  808 FORMAT(8E10.3)
      WRITE(LP,818)
  818 FORMAT('COMFORMING QUALITY INDICES')
      WRITE(LP,808) (CQX(I),CQY(I),I=1,NX)
C
      STOP
      END
C ---------------------------------------------------------------------
C
      SUBROUTINE PTITLE(LR,LW)
C
C   PURPOSE:
C       PRINT OUT THE TITLE OF THE PROBLEM TO BE SOLVED.
C
C   INPUT DATA TO BE READ:
C       0. FLAG(I),I=1,10                        :    10A8
C       1. TITLE(I)                              :    10A8
C
C ---------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION TITLE(10),FLAG(10)
C
      READ(LR,500) (FLAG(I),I=1,10)
      READ(LR,500) (TITLE(I),I=1,10)
  500 FORMAT(10A8)
C
      WRITE(LW,600) (TITLE(I),I=1,10)
  600 FORMAT(1H1,///////////////5X,75('*'),/5X,75('*'),////,15X,
     1 10A8,////5X,75('*'),/5X,75('*'))
C
      RETURN
      END
C ---------------------------------------------------------------------
C
      SUBROUTINE HRMESH(NX,NELX,X,Y,IJK,NODE,MPE,LR,LW)
C
C   PURPOSE:
C       READ THE DATA OF THE FINITE ELEMENT MODEL:  COORDINATES OF
C       NODAL POINTS, ELEMENT CONNECTIVITIES.
C
C   ARGUMENTS:
C       NX     = TOTAL NUMBER OF NODAL POINTS
C       NELX   = TOTAL NUMBER OF FINITE ELEMENTS
C       X,Y    = COORDINATES OF NODAL POINTS
C       IJK    = ELEMENT CONNECTIVITIES
C       NODE   = TOTAL NUMBER OF NODAL POINTS IN AN ELEMENT
C       MPE    = GROUP NUMBER OF THE MATERIALS FOR EACH ELEMENT
C       LR     = CONTROL NUMBER OF THE READ STATEMENT
```

```
C        LW     = CONTROL NUMBER OF THE WRITE STATEMENT
C
C   LOCAL VARIABLES:
C        NODES  = FIRST NODE NUMBER ADDED
C        NELMS  = FIRST ELEMENT NUMBER ADDED
C        NODADD = TOTAL NUMBER OF ADDITIONAL NODAL POINTS
C        NELADD = TOTAL NUMBER OF ADDITIONAL ELEMENTS
C
C   EXAMPLE:
C
C                           N3
C                         / I            IJK(1,NEL)=N1
C                       /   I            IJK(2,NEL)=N2
C                     /     I            IJK(3,NEL)=N3
C                   /   NEL I
C                 /         I
C               N1----------N2
C
C   INPUT DATA TO BE READ:
C        0. FLAG(I),I=1,10                 :    10A8
C        1. NODADD,NELADD                  :    16I5
C        2. II,X(II),Y(II),I=1,NODADD      :    I5,2F10.4
C        3. NN,MPE(NN),(IJK(J,NN),J=1,NODE),
C           N=1,NELADD                     :    5I5
C
C ----------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION X(1),Y(1),IJK(NODE,1),MPE(1),FLAG(10)
C
      READ(LR,598) (FLAG(I),I=1,10)
  598 FORMAT(10A8)
      READ(LR,500) NODADD,NELADD
  500 FORMAT(16I5)
      READ(LR,598) (FLAG(I),I=1,10)
      DO 100 I=1,NODADD
      READ(LR,502) II,X(II),Y(II)
  502 FORMAT(I5,2F10.4)
      IF(II.GT.NX) NX=II
  100 CONTINUE
      READ(LR,598) (FLAG(I),I=1,10)
      DO 102 N=1,NELADD
      READ(LR,500) NN,MPE(NN),(IJK(J,NN),J=1,NODE)
      IF(NN.GT.NELX) NELX=NN
  102 CONTINUE
C
      WRITE(LW,600)
  600 FORMAT(/////10X,'XXXXXXXXXX HRMESH : ELEMENTS AND NODES ',
     1 'XXXXXXXXXX',//15X,
     2 '< ELEMENT CONNECTIVITY >',/)
C
      DO 104 I=1,NELX
  104 WRITE(LW,602) I,MPE(I),(IJK(J,I),J=1,NODE)
  602 FORMAT(10X,I5,3X,I5,5X,'<IJK>',9I5)
      WRITE(LW,604)
  604 FORMAT(////15X,'< COORDINATES >',/)
      WRITE(LW,606) (I,X(I),Y(I),I=1,NX)
  606 FORMAT(5X,I5,3X,'<',2F10.4,1X,'>',5X,I5,3X,'<',2F10.4,
     1 2X,'>')
C
      RETURN
      END
C ----------------------------------------------------------------------
C
      SUBROUTINE MATERL(XYK,LR,LW)
C
C   PURPOSE:
C        READ MATERIAL CONSTANTS KXX,KYY, AND KXY.
C
C   ARGUMENTS:
C        XYK = MATERIAL CONSTANTS IN THE MATRIX FORM
C              XYK(1,N) = KXX IN THE N-TH GROUP
C              XYK(2,N) = KYY IN THE N-TH GROUP
C              XYK(3,N) = KXY IN THE N-TH GROUP
C
C   LOCAL VARIABLES:
C        FLAG = ARRAY FOR THE REMARK
C        MATX = TOTAL NUMBER OF KINDS OF MATERIALS
C
```

```
C   INPUT DATA TO BE READ:
C       0. FLAG(I),I=1,10                    :   10A8
C       1. MATX                              :   16I5
C       2. I,(XYK(J,I),J=1,3),I=1,MATX :   I5,3F10.4
C
C ----------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION XYK(3,1),FLAG(10)
C
      READ(LR,500) (FLAG(I),I=1,10)
  500 FORMAT(10A8)
      READ(LR,502) MATX
  502 FORMAT(16I5)
      READ(LR,504) (I,(XYK(J,I),J=1,3),I=1,MATX)
  504 FORMAT(I5,3F10.4)
C
      WRITE(LW,600) MATX
  600 FORMAT(/////10X,'XXXXXXXXXX MATERIAL CONSTANTS XXXXXXXXXX',//13X,
     1 'NUMBER OF DIFFERENT MATERIALS =',I3,///,10X,
     2 '< MATERIAL CONSTANTS >',/)
      WRITE(LW,602) (I,(XYK(J,I),J=1,3),I=1,MATX)
  602 FORMAT(13X,I3,2X,'KX=',E10.3,2X,'KY=',E10.3,2X,'KXY=',E10.3)
C
      RETURN
      END
C ----------------------------------------------------------------------
C
      SUBROUTINE HBANDW(MBAND,IJK,NELX,NODE,NDF,LW)
C
C   PURPOSE:
C       COMPUTE THE HALF-BAND WIDTH OF THE STIFFNESS MATRIX.
C
C   ARGUMENTS:
C       MBAND = HALF-BAND WIDTH OF THE STIFFNESS MATRIX
C       IJK   = ELEMENT CONNECTIVITIES
C       NELX  = TOTAL NUMBER OF FINITE ELEMENTS
C       NODE  = TOTAL NUMBER OF NODAL POINTS IN AN ELEMENT
C       NDF   = NUMBER OF DEGREES OF FREEDOM PER NODAL POINT
C
C ----------------------------------------------------------------------
C
      DIMENSION IJK(NODE,1)
C
      MBAND=1
      DO 100 NEL=1,NELX
      DO 100 IA=1,NODE
      IJKIA=IJK(IA,NEL)
      DO 100  I=1,NDF
      IJKIAI=NDF*(IJKIA-1)+I
      DO 100 JB=1,NODE
      IJKJB=IJK(JB,NEL)
      DO 100  J=1,NDF
      IJKJBJ=NDF*(IJKJB-1)+J
      IAIJBJ=IJKJBJ-IJKIAI+1
      IF(IAIJBJ.GT.MBAND) MBAND=IAIJBJ
  100 CONTINUE
C
      WRITE(LW,600) MBAND
  600 FORMAT(/////10X,'XXXXXXXXXX BAND WIDTH XXXXXXXXXX',/15X,
     1  'MBAND = ',I8)
C
      RETURN
      END
C ----------------------------------------------------------------------
C
      SUBROUTINE ASEMBO(SK,F,MBAND,NX,NELX,IJK,NODE,NDF,MP,X,Y,XYK,LW)
C
C   PURPOSE:
C       CONSTRUCTS THE GLOBAL STIFFNESS MATRIX AND LOAD VECTOR.
C
C   ARGUMENTS:
C       SK    = GLOBAL STIFFNESS MATRIX
C       F     = GLOBAL LOAD VECTOR
C       MBAND = HALF-BAND WIDTH OF THE STIFFNESS MATRIX
C       NX    = TOTAL NUMBER OF NODAL POINTS
C       NELX  = TOTAL NUMBER OF FINITE ELEMENTS
C       IJK   = ELEMENT CONNECTIVITIES
C       NODE  = TOTAL NUMBER OF NODAL POINTS IN AN ELEMENT
```

```
C       NDF    = NUMBER OF DEGREES OF FREEDOM PER NODAL POINT
C       MP     = GROUP NUMBER OF THE MATERIALS
C       X,Y    = COORDINATES OF NODAL POINTS
C       XYK    = MATERIAL CONSTANTS (KXX,KYY,KXY)
C
C   LOCAL VARIABLES:
C       XE,YE = COORDINATES OF NODAL POINTS IN AN ELEMENT
C       SKE    = ELEMENT STIFFNESS MATRIX
C
C -----------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION SK(1),IJK(NODE,1),MP(1),X(1),Y(1),XYK(3,1),F(1)
      DIMENSION SKE(3,3),FE(3),XE(3),YE(3),XYKE(2,2)
C
C     ( INITIALIZATION )
C
      IMAX=NX*NDF
      DO 100 I=1,IMAX
  100 F(I)=0.DO
      IMAX=MBAND*NX*NDF
      DO 110 I=1,IMAX
  110 SK(I)=0.DO
C
      DO 102 NEL=1,NELX
C
C     ( SET COORDINATES OF NODAL POINTS IN AN ELEMENT )
C
      DO 104 IA=1,NODE
      IJKIA=IJK(IA,NEL)
      XE(IA)=X(IJKIA)
  104 YE(IA)=Y(IJKIA)
C
C     ( SET THE MATERIAL PROPERTIES OF THE ELEMENT )
C
      KKK=MP(NEL)
      XYKE(1,1)=XYK(1,KKK)
      XYKE(1,2)=XYK(3,KKK)
      XYKE(2,1)=XYKE(1,2)
      XYKE(2,2)=XYK(2,KKK)
C
C     ( CONSTRUCT THE ELEMENT STIFFNESS MATRIX )
C
            CALL ESTIF3(NEL,SKE,FE,AREA,XE,YE,XYKE,LW)
C
C     ( ADD THE ELEMENT STIFFNESS MATRIX TO THE GLOBAL ONE )
C
      DO 108 IA=1,NODE
      IJKIA=IJK(IA,NEL)
      DO 108  I=1,NDF
      IAI=NDF*(IA-1)+I
      IJKIAI=NDF*(IJKIA-1)+I
      F(IJKIAI)=F(IJKIAI)+FE(IAI)
      DO 108 JB=1,NODE
      IJKJB=IJK(JB,NEL)
      DO 108  J=1,NDF
      JBJ=NDF*(JB-1)+J
      IJKJBJ=NDF*(IJKJB-1)+J
      IAIJBJ=IJKJBJ-IJKIAI+1
      IF(IAIJBJ.LE.0) GO TO 108
      LOCAT=MBAND*(IJKIAI-1)+IAIJBJ
      SK(LOCAT)=SK(LOCAT)+SKE(IAI,JBJ)
  108 CONTINUE
C
  102 CONTINUE
C
      RETURN
      END
C -----------------------------------------------------------------
C
      SUBROUTINE ESTIF3(NEL,SKE,FE,AREA,XE,YE,XYKE,LW)
C
C   PURPOSE:
C      CONSTRUCT THE ELEMENT STIFFNESS MATRIX BY USING THREE-NODE
C      TRIANGULAR ELEMENT.
C
C   ARGUMENTS:
C       NEL    = ELEMENT NUMBER
```

```
C        SKE    = ELEMENT STIFFNESS MATRIX
C        FE     = ELEMENT LOAD VECTOR
C        AREA   = AREA OF THE TRIANGLE
C        XE,YE  = COORDINATES OF THREE NODAL POINTS IN AN ELEMENT
C        XYKE   = MATERIAL CONSTANTS (KXX,KXY,KYX,KYY)
C
C -----------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION SKE(3,3),FE(3),XE(3),YE(3),DN(3,2),XYKE(2,2)
C
C     ( INITIALIZATION )
C
      DO 200 I=1,3
      FE(I)=0.DO
      DO 200 J=1,3
  200 SKE(I,J)=0.DO
C
C     ( COMPUTE THE FIRST GRADIENT OF THE SHAPE FUNCTIONS )
C
            CALL GRADI3(NEL,XE,YE,DN,AREA,LW)
C
C     ( CONSTRUCT THE STIFFNESS MATRIX )
C
      DO 100 I=1,3
      DO 100 J=1,3
      SKEIJ=0.
      DO 102 K=1,2
      DO 102 L=1,2
  102 SKEIJ=SKEIJ+XYKE(K,L)*DN(I,K)*DN(J,L)
  100 SKE(I,J)=SKEIJ*AREA
C
      RETURN
      END
C -----------------------------------------------------------------------
C
      SUBROUTINE GRADI3(NEL,XE,YE,DN,AREA,LW)
C
C   PURPOSE:
C       CONSTRUCT THE FIRST DERIVATIVES OF THE SHAPE FUNCTIONS.
C
C   ARGUMENTS:
C       NEL    = ELEMENT NUMBER
C       XE,YE  = COORDINATES OF NODAL POINTS IN AN ELEMENT
C       DN     = FIRST DERIVATIVES OF THE SHAPE FUNCTIONS
C       AREA   = AREA OF THE TRIANGLE
C
C -----------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION XE(3),YE(3),DN(3,2)
C
      DET=XE(2)*(YE(3)-YE(1))+XE(3)*(YE(1)-YE(2))+XE(1)*(YE(2)-YE(3))
      AREA=0.5DO*DET
      IF(DET.LE.1.E-30) GO TO 100
C
      DN(1,1)=(YE(2)-YE(3))/DET
      DN(2,1)=(YE(3)-YE(1))/DET
      DN(3,1)=(YE(1)-YE(2))/DET
      DN(1,2)=(XE(3)-XE(2))/DET
      DN(2,2)=(XE(1)-XE(3))/DET
      DN(3,2)=(XE(2)-XE(1))/DET
C
      RETURN
C
  100 WRITE(LW,600) NEL, (XE(I),YE(I),I=1,3)
  600 FORMAT(///10X,'******** DETERMINANT IS LESS THAN ZERO ********',
     1 /,15X,I5,5X,3('<',2F8.3,'>'))
      STOP
C
      END
C -----------------------------------------------------------------------
C
      SUBROUTINE HEATGE(F,NX,NELX,IJK,X,Y,LR,LW)
C
C   PURPOSE:
C       CONSTRUCT THE GENERALIZED LOAD VECTOR DUE TO THE HEAT
C       GENERATION.
```

```
C
C   ARGUMENTS:
C       F    = GENERALIZED LOAD VECTOR
C       NX   = TOTAL NUMBER OF NODAL POINTS
C       NELX = TOTAL NUMBER OF FINITE ELEMENTS
C       IJK  = ELEMENT CONNECTIVITIES
C       X,Y  = COORDINATES OF NODAL POINTS
C
C   LOCAL VARIABLES:
C       NHEATG = TOTAL NUMBER OF TERMS OF POLYNOMIALS
C              = 0  IF THERE ARE NO HEAT SOURCES/SINKS
C              = NEGATIVE   IF POINT SOURCES/SINKS EXIST.  IN THIS CASE
C                -NHEATG BECOMES THE NUMBER OF SOURCES/SINKS
C       CXY    = COEFFICIENTS OF THE POLYNOMIAL BASIS
C       XE,YE = COORDINATES OF NODAL POINTS IN AN ELEMENT
C       XC,YC = COORDINATES OF THE CENTROID OF AN ELEMENT
C       AREA  = AREA OF AN ELEMENT
C
C   DIMENSIONING:
C       CXY(M,M)   M >= NHEATG
C
C   NOTE:
C       M = 11  AND  8 >= NHEATG  ARE ASSUMED IN THIS ROUTINE.
C
C   INPUT DATA TO BE READ:
C       0. FLAG(I),I=1,10                  :   10A8
C       1. NHEATG                          :   16I5
C       IF(NHEATG.LT.0) GO TO 3
C       DO 100 I=1,NHEATG
C       2. CXY(I,J),J=1,NHEATG             :   8F10.4
C       100 CONTINUE
C       RETURN
C       3. NODE,VALUE,I=1,-NHEATG          :   I5,F10.4
C
C ---------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION F(1),IJK(3,1),X(1),Y(1),XE(3),YE(3),CXY(11,11),
     1          FLAG(10)
C
C     ( READ INPUT DATA FOR THE BODY FORCE )
C
      READ(LR,504) (FLAG(I),I=1,10)
  504 FORMAT(10A8)
      READ(LR,500) NHEATG
  500 FORMAT(16I5)
      IF(NHEATG.LE.0) GO TO 300
C
C     ( DISTRIBUTED HEAT SOURCE/SINK )
C
      DO 100 I=1,NHEATG
  100 READ(LR,502) (CXY(I,J),J=1,NHEATG)
  502 FORMAT(8F10.4)
C
      DO 102 NEL=1,NELX
C
C     ( SET AREA AND COORDINATES OF THE CENTROID )
C
      DO 104 I=1,3
      IJKI=IJK(I,NEL)
      XE(I)=X(IJKI)
  104 YE(I)=Y(IJKI)
      DET=XE(2)*(YE(3)-YE(1))+XE(3)*(YE(1)-YE(2))+XE(1)*(YE(2)-YE(3))
      AREA=0.5D0*DET
      XC=(XE(1)+XE(2)+XE(3))/3.
      YC=(YE(1)+YE(2)+YE(3))/3.
C
C     ( COMPUTE THE BODY FORCE )
C
      FNEL=0.
      DO 106 I=1,NHEATG
      I1=I-1
      DO 106 J=1,NHEATG
      J1=J-1
  106 FNEL=FNEL+CXY(I,J)*(XC**I1)*(YC**J1)
C
      DO 108 I=1,3
      IJKI=IJK(I,NEL)
```

```
  108 F(IJKI)=F(IJKI)+FNEL*AREA/3.
C
  102 CONTINUE
      WRITE(LW,600)
  600 FORMAT(/////10X,'XXXXXXXXXX BODY FORCES XXXXXXXXXX',//15X,
     1 '< COEFFECIENTS  CXY  >',/)
      DO 110 I=1,NHEATG
  110 WRITE(LW,602) (CXY(I,J),J=1,NHEATG)
  602 FORMAT(15X,10F8.3)
      GO TO 304
C
C     ( POINT HEAT SOURCES/SINKS )
C
  300 IF(NHEATG.EQ.0) RETURN
      NHEATG=-NHEATG
      DO 302 I=1,NHEATG
      READ(LR,530) NODE,VALUE
  530 FORMAT(I5,F10.4)
  302 F(NODE)=F(NODE)+VALUE
C
  304 WRITE(LW,604)
  604 FORMAT(//10X,'< LOAD VECTOR DUE TO HEAT GENERATION >',/)
      WRITE(LW,606) (I,F(I),I=1,NX)
  606 FORMAT(5X,I4,1X,E10.3,3X,I4,1X,E10.3,3X,I4,1X,E10.3)
C
      RETURN
      END
C -----------------------------------------------------------------------
C
      SUBROUTINE BOUND1(MBAND,SK,F,LR,LW)
C
C  PURPOSE:
C      READ THE 1-ST BOUNDARY CONDITION AND MODIFY THE STIFFNESS
C      MATRIX AND THE LOAD VECTOR USING PENALTY METHODS.
C
C  ARGUMENTS:
C      MBAND = HALF-BAND WIDTH OF THE STIFFNESS MATRIX
C      SK    = STIFFNESS MATRIX
C      F     = LOAD VECTOR
C
C  LOCAL VARIABLES:
C      NBX   = TOTAL NUMBER OF NODAL POINTS ON WHICH 1-ST BOUNDARY
C              CONDITION IS APPLIED
C      PENALT= PENALTY PARAMETER  (TO BE A SUFFICIENTLY LARGE NUMBER )
C              10000.*MAX(XK,YK) IS RECOMMENDED
C      XK,YK = HEAT CONDUCTIVITY IN THE X AND Y DIRECTIONS
C      NODE  = NODE NUMBER ON WHICH BOUNDARY CONDITION IS GIVEN
C      VALUE = BOUNDARY VALUE SPECIFIED
C
C  INPUT DATA TO BE READ:
C      0. FLAG(I),I=1,10           :    10A8
C      1. NBX,PENALT               :    I5,F20.5
C      2. NODE,VALUE,I=1,NBX       :    I5,F10.4
C
C -----------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION SK(1),F(1),FLAG(10)
C
C     ( READ INPUT DATA )
C
      READ(LR,504) (FLAG(I),I=1,10)
  504 FORMAT(10A8)
      READ(LR,500) NBX,PENALT
  500 FORMAT(I5,F20.5)
      WRITE(LW,600) PENALT
  600 FORMAT(/////10X,'XXXXXXXXXX 1-ST BOUNDARY CONDITION XXXXXXXXXX',/
     1 15X,'PENALTY PARAMETER =',E10.3,//)
C
      DO 100 I=1,NBX
      READ(LR,502) NODE,VALUE
  502 FORMAT(I5,F10.4)
      WRITE(LW,602) I,NODE,VALUE
  602 FORMAT(15X,'<',I3,'>',4X,I4,3X,F10.4)
C
      SK(MBAND*(NODE-1)+1)=SK(MBAND*(NODE-1)+1)+PENALT
      F(NODE)=F(NODE)+PENALT*VALUE
C
```

```
  100 CONTINUE
C
      RETURN
      END
C ---------------------------------------------------------------------
C
      SUBROUTINE BOUND2(F,X,Y,LR,LW)
C
C   PURPOSE:
C       READ THE 2-ND BOUNDARY CONDITION AND FORM THE LOAD VECTOR.
C        <  DU/DN = H   ON  S2  >
C
C   ARGUMENTS:
C       F    = GENRALIZED LOAD VECTOR
C       X,Y = COORDINATES OF NODAL POINTS
C
C   LOCAL VARIABLES:
C       FLAG  = REMARK FOR THE DATA
C       NBX   = TOTAL NUMBER OF LINE ELEMENTS ON THIS BOUNDARY
C       NODE1 = NUMBER OF THE LEFT NODAL POINT IN A LINE ELEMENT
C       NODE2 = NUMBER OF THE RIGHT NODAL POINT IN A LINE ELEMENT
C       VALUE = VALUE OF THE FLUX FROM THE BOUNDARY
C
C   INPUT DATA TO BE READ:
C       0. FLAG(I),I=1,10                      :   10A8
C       1. NBX                                 :   16I5
C       2. NODE1,NODE2,VALUE,I=1,NBX           :   2I5,F10.4
C
C ---------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION FLAG(10),F(1),X(1),Y(1)
C
      READ(LR,500) (FLAG(I),I=1,10)
  500 FORMAT(10A8)
      READ(LR,502) NBX
  502 FORMAT(16I5)
      IF(NBX.LE.0) RETURN
      WRITE(LW,600)
  600 FORMAT(/////10X,'XXXXXXXXXX 2-ND BOUNDARY CONDITION XXXXXXXXXX',
     1 //15X,29('-'),/15X,' NODE I     VALUE   ',/15X,29('-'))
C
      DO 100 I=1,NBX
      READ(LR,504) NODE1,NODE2,VALUE
  504 FORMAT(2I5,F10.4)
      WRITE(LW,602) NODE1,NODE2,VALUE
  602 FORMAT(15X,2I5,' I ',E10.3)
      DH=DSQRT((X(NODE1)-X(NODE2))**2+(Y(NODE1)-Y(NODE2))**2)
      F(NODE1)=F(NODE1)+0.5D0*VALUE*DH
      F(NODE2)=F(NODE2)+0.5D0*VALUE*DH
  100 CONTINUE
C
      WRITE(LW,604)
  604 FORMAT(15X,29('-'))
C
      RETURN
      END
C ---------------------------------------------------------------------
C
      SUBROUTINE BOUND3(MBAND,SK,F,X,Y,IJB,MPB,XKB,U0,LR,LW)
C
C   PURPOSE:
C       READ THE 3-RD BOUNDARY CONDITION AND MODIFY THE STIFFNESS
C       MATRIX AND THE LOAD VECTOR.
C       < DU/DN = -K( U - U0 )  ON  S3  >
C
C   ARGUMENTS:
C       MBAND = HALF-BAND WIDTH OF THE STIFFNESS MATRIX
C       SK    = GLOBAL STIFFNESS MATRIX
C       F     = LOAD VECTOR
C       X,Y   = COORDINATES OF THE NODAL POINTS
C       IJB   = ELEMENT CONNECTIVITY ON THE BOUNDARY
C       MPB   = GROUP NUMBER OF THE MATERIALS ON THE BOUNDARY
C       XKB   = CONSTANTS  "K"  ON THE BOUNDARY
C       U0    = CONSTANTS  "U0"  ON THE BOUNDARY
C
C   LOCAL VARIABLES:
C       FLAG   = REMARK TO THE INPUT DATA
```

```
C        NBLX    = TOTAL NUMBER OF BOUNDARY ELEMENTS
C        MPBX    = TOTAL NUMBER OF MATERIALS ON THE BOUNDARY
C
C   INPUT DATA TO BE READ:
C        0. FLAG(I),I=1,10                           :    10A8
C        1. NBLX,MPBX                                :    16I5
C        2. I,MPB(I),(IJB(J,I),J=1,2),I=1,NBLX       :    4I5
C        3  I,XKB(I),U0(I),I=1,MPBX                  :    I5,2F10.4
C
C -----------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION SK(1),F(1),X(1),Y(1),IJB(2,1),MPB(1),XKB(1),U0(1)
      DIMENSION SKE(2,2),FE(2),FLAG(10)
C
C     ( INPUT DATA )
C
      READ(LR,500) (FLAG(I),I=1,10)
  500 FORMAT(10A8)
      READ(LR,502) NBLX,MPBX
  502 FORMAT(16I5)
      IF(NBLX.LE.0) RETURN
      READ(LR,504) (I,MPB(I),(IJB(J,I),J=1,2),I=1,NBLX)
  504 FORMAT(4I5)
      READ(LR,506) (I,XKB(I),U0(I),I=1,MPBX)
  506 FORMAT(I5,2F10.4)
C
      WRITE(LW,600)
  600 FORMAT(/////10X,'XXXXXXXXXX 3-RD BOUNDARY CONDITION XXXXXXXXXX',
     1 //10X,'----- BOUNDARY ELEMENT CONNECTIVITY -----',/)
      WRITE(LW,602) (I,MPB(I),(IJB(J,I),J=1,2),I=1,NBLX)
  602 FORMAT(2(5X,I5,3X,'<MPB>',I3,5X,'<IJB>',2I5))
      WRITE(LW,604)
  604 FORMAT(///10X,'----- MATERIAL CONSTANTS ON THE BOUNDARY -----',
     1 /)
      WRITE(LW,606) (I,XKB(I),U0(I),I=1,MPBX)
  606 FORMAT(10X,I5,3X,'K=',E10.3,2X,'U0=',E10.3)
C
C     ( FOR STIFFNESS MATRIX AND LOAD VECTOR )
C
      DO 100 NEL=1,NBLX
C
C     ( SET VARIABLES )
C
      N1=IJB(1,NEL)
      N2=IJB(2,NEL)
      X1=X(N1)
      Y1=Y(N1)
      X2=X(N2)
      Y2=Y(N2)
      HR=DSQRT((X2-X1)**2+(Y2-Y1)**2)
      KKK=MPB(NEL)
      XKBE=XKB(KKK)
      U0E=U0(KKK)
C
C     ( ELEMENT STIFFNESS MATRIX )
C
      SKE(1,1)=XKBE*HR/3.D0
      SKE(2,2)=SKE(1,1)
      SKE(1,2)=XKBE*HR/6.D0
      SKE(2,1)=SKE(1,2)
C
C     ( ELEMENT LOAD VECTOR )
C
      FE(1)=XKBE*U0E*HR/2.D0
      FE(2)=FE(1)
C
C     ( ASSEMBLING : BAND SOLVER )
C
      DO 102 I=1,2
      NI=IJB(I,NEL)
      F(NI)=F(NI)+FE(I)
      DO 102 J=1,2
      NJ=IJB(J,NEL)
      KI=NJ-NI+1
      IF(KI.LE.0) GO TO 102
      SK(MBAND*(NI-1)+KI)=SK(MBAND*(NI-1)+KI)+SKE(I,J)
  102 CONTINUE
```

```
C
  100 CONTINUE
C
      RETURN
      END
C ------------------------------------------------------------------------
C
      SUBROUTINE BANDSL(AK,R,NEQ,MBAND,KKK)
C
C   PURPOSE:
C       SOLVES A SYSTEM OF LINEAR EQUATIONS.
C       ( BANDED SYMMETRIC COEFFICIENT MATRIX )
C
C   CONTROL NUMBERS:
C       KKK = 0   BAND SOLVER
C       KKK = 1   TRIANGULATION
C       KKK = 2   REDUCTION AND BACK SUBSTITUTION
C
C   ARGUMENTS:
C       AK    = BANDED SYMMETRIC MATRIX TO BE SOLVED
C       R     = RIGHT HAND SIDE ( LOAD VECTOR )
C       NEQ   = TOTAL NUMBER OF DEGREES OF FREEDOM
C       MBAND = HALF-WIDTH OF BAND
C       KKK   = CONTROL NUMBER OF THE FLOW OF THE ROUTINE
C
C   NOTE:
C       HOW TO STORE THE STIFFNESS MATRIX
C
C         ( K11 K12 K13     K15         )
C         I     K22 K23 K24 K25         I
C         I         K33 K34 K35 K36   O I
C         I   SMY       K44 K45 K46 K47 I
C         I                 K55 K56 K57 I
C         I                     K66 K67 I
C         (                         K77 )
C
C       BANDED STIFFNESS MATRIX
C
C         ( K11 K12 K13   O K15 )
C         I K22 K23 K24 K25   O I
C         I K33 K34 K35 K36   O I
C         I K44 K45 K46 K47   O I
C         I K55 K56 K57   O   O I
C         I K66 K67   O   O   O I
C         ( K77   O   O   O   O )
C
C          NEQ = 7      MBAND = 5
C
C       ONE-DIMENSIONAL ARRAY OF THE STIFFNESS MATRIX "AK"
C
C       AK/K11,K12,K13,O,K15,K22,K23,K24,K25,O,K33,K34,K35,K36,O,
C          K44,K45,K46,K47,O,K55,K56,K57,O,O,K66,K67,O,O,O,K77,O,O,O,O
C
C ------------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION AK(1),R(1)
C
      NRS=NEQ-1
      NR=NEQ
      IF(KKK.EQ.2) GO TO 200
      DO 120 N=1,NRS
      M=N-1
      MR=MINO(MBAND,NR-M)
      PIVOT=AK(MBAND*(N-1)+1)
      DO 120 L=2,MR
      CP=AK(MBAND*(N-1)+L)/PIVOT
      IF(CP) 130,120,130
  130 I=M+L
      J=0
      DO 110 K=L,MR
      J=J+1
      LOCAT=MBAND*(I-1)+J
  110 AK(LOCAT)=AK(LOCAT)-CP*AK(MBAND*(N-1)+K)
  120 AK(MBAND*(N-1)+L)=CP
      IF(KKK.EQ.1) RETURN
  200 DO 230 N=1,NRS
      M=N-1
```

```
      MR=MINO(MBAND,NR-M)
      CP=R(N)
      IF(CP)210,230,210
  210 R(N)=CP/AK(MBAND*(N-1)+1)
      DO 220 L=2,MR
      I=M+L
  220 R(I)=R(I)-AK(MBAND*(N-1)+L)*CP
  230 CONTINUE
      R(NR)=R(NR)/AK(MBAND*(NR-1)+1)
      DO 320 I=1,NRS
      N=NR-I
      M=N-1
      MR=MINO(MBAND,NR-M)
      DO 320 K=2,MR
      L=M+K
  320 R(N)=R(N)-AK(MBAND*(N-1)+K)*R(L)
C
      RETURN
      END
C ---------------------------------------------------------------------
C
      SUBROUTINE OUTPUT(F,P,NX,NELX,X,Y,IJK,MP,XYK,LW)
C
C   PURPOSE:
C       OUTPUTS THE TEMPERATURE AND THE HEAT FLUX VECTORS.
C
C   ARGUMENTS:
C       F    = SOLUTION OBTAINED BY THE F.E. ANALYSIS
C       P    = STORES THE HEAT FLUX VECTORS AND QUALITY INDICES
C       NX   = TOTAL NUMBER OF NODAL POINTS
C       NELX = TOTAL NUMBER OF FINITE ELEMENTS
C       X,Y  = COORDINATES OF NODAL POINTS
C       IJK  = ELEMENT CONNECTIVITIES
C       MP   = GROUP NUMBER OF THE MATERIALS
C       XYK  = MATERIAL CONSTANT IN THE MATRIX FORM
C
C   LOCAL VARIABLES:
C       DFX,DFY = FIRST DERIVATIVES OF THE TEMPERATURE
C       QX,QY   = COMPONENTS OF THE HEAT FLUX VECTOR
C
C ---------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION F(1),P(4,1),IJK(3,1),X(1),Y(1),XYK(3,1),MP(1)
      DIMENSION XE(3),YE(3),FE(3),DN(3,2)
C
C     ( OUTPUT THE SOLUTION VECTOR )
C
       WRITE(LW,600)
  600 FORMAT(1H1,///////////////,5X,70('*'),/5X,70('*'),///20X,
     1 'THE SOLUTION VECTORS',///5X,70('*'),/5X,70('*'))
      WRITE(LW,602)
  602 FORMAT(1H1,//10X,43('-'),/10X,' NODE I     X         Y     ',
     1 'I     F',/10X,43('-'))
      WRITE(LW,604) (I,X(I),Y(I),F(I),I=1,NX)
  604 FORMAT(10X,I4,2X,'I',1X,F8.3,F9.3,2X,'I',1X,E12.5)
      WRITE(LW,606)
  606 FORMAT(10X,43('-'))
C
C     ( OUTPUT THE GRADIENT )
C
      WRITE(LW,608)
  608 FORMAT(1H1,//////////5X,70('*'),/5X,70('*'),///20X,
     1 ' THE GRADIENTS OF THE SOLUTION',///5X,70('*'),/5X,70('*'))
      WRITE(LW,610)
  610 FORMAT(1H1,//5X,62('-'),/5X,' NEL I     X         Y     ',
     1 'I       F     I     FX         FY',/5X,62('-'))
C
      DO 100 NEL=1,NELX
C
C     ( SET THE COORDINATES )
C
      DO 102 I=1,3
      IJKI=IJK(I,NEL)
      XE(I)=X(IJKI)
      YE(I)=Y(IJKI)
  102 FE(I)=F(IJKI)
C
```

```
      XC=(XE(1)+XE(2)+XE(3))/3.DO
      YC=(YE(1)+YE(2)+YE(3))/3.DO
      FC=(FE(1)+FE(2)+FE(3))/3.DO
C
C     ( CALL THE GRADIENT ROUTINE )
C
            CALL GRADI3(NEL,XE,YE,DN,AREA,LW)
C
C     ( GRADIENT OF THE SOLUTION )
C
      KK=MP(NEL)
      XKNEL=XYK(1,KK)
      YKNEL=XYK(2,KK)
      XYKNEL=XYK(3,KK)
      DFX=0.DO
      DFY=0.DO
      DO 104 I=1,3
      DFX=DFX+FE(I)*DN(I,1)
  104 DFY=DFY+FE(I)*DN(I,2)
      QX=-(XKNEL*DFX+XYKNEL*DFY)
      QY=-(XYKNEL*DFX+YKNEL*DFY)
      P(1,NEL)=QX
      P(2,NEL)=QY
C
      WRITE(LW,620) NEL,XC,YC,FC,QX,QY
  620 FORMAT(5X,I4,' I ',F8.3,F9.3,' I ',E10.3,' I ',2E10.3)
C
  100 CONTINUE
      WRITE(LW,612)
  612 FORMAT(5X,62('-'))
C
      RETURN
      END
C -------------------------------------------------------------------
C
      SUBROUTINE QUALIT(NELX,NX,MBAND,IJK,X,Y,P,SK,CQX,CQY,NODE,LW)
C
C   PURPOSE :
C       COMPUT QUALTY INDICES OF THE FINITE ELEMENT APPROXIMATION.
C
C   ARGUMENTS :
C       NELX   = TOTAL NUMBER OF ELEMENTS
C       NX     = TOTAL NUMBER OF NODAL POINTS
C       MBAND  = HALF BAND WIDTH OF THE MASS MATRIX
C       IJK    = ELEMENT CONNECTIVITIES
C       X,Y    = COORDINATES OF NODAL POINTS
C       P(1,N) = X COMPONENT OF THE HEAT FLUX IN THE N TH ELEMENT
C       P(2,N) = Y COMPONENT OF THE HEAT FLUX IN THE N TH ELEMENT
C       P(3,N) = QUALITY INDEX IN THE N TH ELEMENT (TEMPERATURE)
C       P(4,N) = QUALITY INDEX IN THE N TH ELEMENT (HEAT FLUX)
C       SK     = MASS MATRIX FOR THE LEAST SQUARES METHOD
C       CQX    = QUALITY INDEX AT NODAL POINTS (TEMPERATURE)
C       CQY    = QUALITY INDEX AT NODAL POINTS (HEAT FLUX)
C       NODE   = TOTAL NUMBER OF NODAL POINTS IN AN ELEMENT
C
C   LOCAL VARIABLES:
C       HX    = SIZE OF FINITE ELEMENT IN THE X DIRECTION
C       HY    = SIZE OF FINITE ELEMENT IN THE Y DIRECTION
C       QI0AV = AVERAGE QUALITY INDEX FOR TEMPERATURE
C       QI1AV = AVERAGE QUALITY INDEX FOR HEAT FLUX
C
C   NOTE:
C       ARRAYS "CQX" AND "CQY" ARE USED FOR THE CONFORMING HEAT FLUX
C       VECTORS FIRST, THEN STORE THE QUALITY INDICES AT NODAL POINTS.
C
C -------------------------------------------------------------------
C
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION IJK(NODE,1),P(4,1),CQX(1),CQY(1),X(1),Y(1),SK(1)
      DIMENSION SME(3,3)
C
      DATA SME/2.DO,1.DO,1.DO,1.DO,2.DO,1.DO,1.DO,1.DO,2.DO/
C
C     ( INITIAL SET )
C
      IMAX=MBAND*NX
      DO 100 I=1,IMAX
  100 SK(I)=0.DO
      DO 102 I=1,NX
```

```
      CQX(I)=0.D0
  102 CQY(I)=0.D0
C
C     ( LEAST SQUARES METHODS )
C
      DO 200 NEL=1,NELX
C
      IJK1=IJK(1,NEL)
      IJK2=IJK(2,NEL)
      IJK3=IJK(3,NEL)
      X1=X(IJK1)
      X2=X(IJK2)
      X3=X(IJK3)
      Y1=Y(IJK1)
      Y2=Y(IJK2)
      Y3=Y(IJK3)
      AREA=0.5D0*(X1*(Y2-Y3)+X2*(Y3-Y1)+X3*(Y1-Y2))
      DQX=P(1,NEL)
      DQY=P(2,NEL)
C
      DO 202 IA=1,NODE
      IJKIA=IJK(IA,NEL)
      CQX(IJKIA)=CQX(IJKIA)+DQX*AREA/3.D0
      CQY(IJKIA)=CQY(IJKIA)+DQY*AREA/3.D0
      DO 202 JB=1,NODE
      IJKJB=IJK(JB,NEL)
      IAJB=IJKJB-IJKIA
      IF(IAJB.LT.0) GO TO 202
      ND=MBAND*(IJKIA-1)+IAJB+1
      SK(ND)=SK(ND)+SME(IA,JB)*AREA/12.D0
  202 CONTINUE
C
  200 CONTINUE
C
C     ( SOLVE THE SYSTEMS OF LINEAR EQUATIONS )
C
            CALL BANDSL(SK,CQX,NX,MBAND,0)
C
            CALL BANDSL(SK,CQY,NX,MBAND,2)
C
      WRITE(LW,600)
  600 FORMAT(/////5X,70('*'),//10X,
     1 'CONFORMING HEAT FLUX VECTORS AT NODES',//5X,70('*'),/)
      WRITE(LW,602)
  602 FORMAT(5X,35('-'),/5X,' NODE I      CQX      I      CQY',
     1 /5X,35('-'))
      WRITE(LW,604) (I,CQX(I),CQY(I),I=1,NX)
  604 FORMAT(5X,I5,' I ',E11.4,' I ',E11.4)
      WRITE(LW,606)
  606 FORMAT(5X,35('-'))
C
C     ( COMPUTE THE QUALITY INDEX  QI0 AND QI1 )
C
      QI0TL=0.D0
      QI1TL=0.D0
      QI0MAX=-1.D20
      QI1MAX=-1.D20
C
      DO 300 NEL=1,NELX
C
      IJK1=IJK(1,NEL)
      IJK2=IJK(2,NEL)
      IJK3=IJK(3,NEL)
      X1=X(IJK1)
      X2=X(IJK2)
      X3=X(IJK3)
      Y1=Y(IJK1)
      Y2=Y(IJK2)
      Y3=Y(IJK3)
      DET=X1*(Y2-Y3)+X2*(Y3-Y1)+X3*(Y1-Y2)
      AREA=0.5D0*DET
C
      DQX=DABS(P(1,NEL))
      DQY=DABS(P(2,NEL))
      HX=DMAX1(DABS(X1-X2),DABS(X2-X3),DABS(X3-X1))
      HY=DMAX1(DABS(Y1-Y2),DABS(Y2-Y3),DABS(Y3-Y1))
      QI0XX=HX*DQX
      QI0YY=HY*DQY
      QI0NEL=DMAX1(QI0XX,QI0YY)*DSQRT(AREA)
```

```
      P(3,NEL)=QIONEL
      QIOTL=QIOTL+QIONEL
      IF(QIOMAX.LT.QIONEL) QIOMAX=QIONEL
C
      B1=(Y2-Y3)/DET
      B2=(Y3-Y1)/DET
      B3=(Y1-Y2)/DET
      C1=(X3-X2)/DET
      C2=(X1-X3)/DET
      C3=(X2-X1)/DET
      DQXX=CQX(IJK1)*B1+CQX(IJK2)*B2+CQX(IJK3)*B3
      DQXY=CQX(IJK1)*C1+CQX(IJK2)*C2+CQX(IJK3)*C3
      DQYX=CQY(IJK1)*B1+CQY(IJK2)*B2+CQY(IJK3)*B3
      DQYY=CQY(IJK1)*C1+CQY(IJK2)*C2+CQY(IJK3)*C3
      QI1XXX=HX*DABS(DQXX)
      QI1XYX=HX*DABS(DQYX)
      QI1YXY=HY*DABS(DQXY)
      QI1YYY=HY*DABS(DQYY)
      QI1NEL=DMAX1(QI1XXX,QI1XYX,QI1YXY,QI1YYY)*DSQRT(AREA)
      P(4,NEL)=QI1NEL
      QI1TL=QI1TL+QI1NEL
      IF(QI1MAX.LT.QI1NEL) QI1MAX=QI1NEL
C
  300 CONTINUE
C
      QIOAV=QIOTL/NELX
      QI1AV=QI1TL/NELX
C
      WRITE(LW,620) QIOAV,QI1AV
  620 FORMAT(/////5X,70('*'),//15X,
     1 'QUALITY INDICES FOR TEMPERATURE AND HEAT FLUX',//5X,70('*'),
     2 //5X,'AVERAGE QUALITY INDEX (TEMPERATURE) = ',E12.5,/5X,
     3 'AVERAGE QUALITY INDEX (HEAT FLUX) = ',E12.5,//,
     4 5X,45('-'),/5X,' NEL   I  LO L1  I     QIO      I      QI1',
     5 /5X,45('-'))
      DO 310 NEL=1,NELX
      QIONEL=P(3,NEL)
      QI1NEL=P(4,NEL)
      LO=10*(QIONEL+1.D-10)/QIOMAX
      L1=10*(QI1NEL+1.D-10)/QI1MAX
      WRITE(LW,622) NEL,LO,L1,QIONEL,QI1NEL
  622 FORMAT(5X,I5,' I ',2I3,' I ',E11.4,' I ',E11.4)
  310 CONTINUE
      WRITE(LW,626)
  626 FORMAT(5X,45('-'))
C
C     ( COMPUTE NODAL VALUES OF QUALITY INDICES )
C
      DO 400 I=1,NX
      CQX(I)=0.DO
  400 CQY(I)=0.DO
C
      DO 402 NEL=1,NELX
      IJK1=IJK(1,NEL)
      IJK2=IJK(2,NEL)
      IJK3=IJK(3,NEL)
      X1=X(IJK1)
      X2=X(IJK2)
      X3=X(IJK3)
      Y1=Y(IJK1)
      Y2=Y(IJK2)
      Y3=Y(IJK3)
      AREA=0.5DO*(X1*(Y2-Y3)+X2*(Y3-Y1)+X3*(Y1-Y2))
      CQX(IJK1)=CQX(IJK1)+P(3,NEL)*AREA/3.DO
      CQX(IJK2)=CQX(IJK2)+P(3,NEL)*AREA/3.DO
      CQX(IJK3)=CQX(IJK3)+P(3,NEL)*AREA/3.DO
      CQY(IJK1)=CQY(IJK1)+P(4,NEL)*AREA/3.DO
      CQY(IJK2)=CQY(IJK2)+P(4,NEL)*AREA/3.DO
      CQY(IJK3)=CQY(IJK3)+P(4,NEL)*AREA/3.DO
  402 CONTINUE
C
            CALL BANDSL(SK,CQX,NX,MBAND,2)
C
            CALL BANDSL(SK,CQY,NX,MBAND,2)
C
      RETURN
      END
```